column_drop_add_with_default.sql

SQL Server 2005 has a number of built-in templates for scripting various things, and you can easily create your own.

For example, the following template will drop a column if it exists (even if it has a default constraint) and recreate the column (with a default constraint) in an existing table:

DECLARE @constraint sysname, @sql nvarchar(4000)

IF EXISTS (SELECT 1 FROM sys.columns WHERE object_id = OBJECT_ID(N'dbo.<table_name, sysname, table_name>') AND [name] = '<column_name, sysname, column_name>')

BEGIN


SELECT @constraint = [name] FROM sys.default_constraints


WHERE parent_object_id = (SELECT object_id FROM sys.tables WHERE [name] = '<table_name, sysname, table_name>')


AND parent_column_id = (SELECT column_id FROM sys.columns WHERE [name] = '<column_name, sysname, column_name>' AND object_id = OBJECT_ID(N'dbo.<table_name, sysname, table_name>'))


IF @constraint IS NOT NULL


BEGIN



SET @sql = 'ALTER TABLE dbo.<table_name, sysname, table_name> DROP CONSTRAINT '+ @constraint



EXEC sp_executesql @sql


END


EXEC sp_executesql N'ALTER TABLE <table_name, sysname, table_name> DROP COLUMN <column_name, sysname, column_name>'

END

GO

ALTER TABLE dbo.<table_name, sysname, table_name> ADD <column_name, sysname, column_name> <data_type, nvarchar, bit> <nullable, nvarchar, NOT NULL> CONSTRAINT DF__<table_name, sysname, table_name>__<column_name, sysname, column_name> DEFAULT (<default_value, nvarchar, 0>)

GO

When you open the template file in SSMS you are prompted to connect to the database, this is normal, regardless of any windows that are open & connected. Note: It will always connect to master, so remember to select the correct db before you actually execute the resulting script.

In this example, we have an existing table Companies with a column named Flag (with or without a default) & we are replacing it with a non-nullable bit column Flag with a default value of 0.

Why do we drop and then recreate the column? So this script can be run and re-run without generating any error messages. In this way, the script can be part of a large batch file that can be built in development, then applied to QA and then finally production.

Press Ctrl-Shift-M to display a popup screen where you enter the desired values, then click OK. 

The popup screen looks like this:

The script will be filled-in as follows:

DECLARE @constraint sysname, @sql nvarchar(4000)

IF EXISTS (SELECT 1 FROM sys.columns WHERE object_id = OBJECT_ID(N'dbo.Companies') AND [name] = 'Flag')

BEGIN


SELECT @constraint = [name] FROM sys.default_constraints


WHERE parent_object_id = (SELECT object_id FROM sys.tables WHERE [name] = 'Companies')


AND parent_column_id = (SELECT column_id FROM sys.columns WHERE [name] = 'Flag' AND object_id = OBJECT_ID(N'dbo.Companies'))


IF @constraint IS NOT NULL


BEGIN



SET @sql = 'ALTER TABLE dbo.Companies DROP CONSTRAINT '+ @constraint



EXEC sp_executesql @sql


END


EXEC sp_executesql N'ALTER TABLE Companies DROP COLUMN Flag'

END

GO

ALTER TABLE dbo.Companies ADD Flag bit NOT NULL CONSTRAINT DF__Companies__Flag DEFAULT (0)

GO
The resulting script does several things:

· It tests that the named column exists

· If the column exists, it looks up the name of any default constraint

· If a default constraint exists, it drops the constraint

· If the column exists, it drops the column

· Finally, it adds the column with the constraint

When you create your own templates, remember to change the attributes to Read-Only, because after you fill-in the values the file name does not change, so you don’t want to overwrite your custom template.

